


Задача 6

(Задача 32.26. Задачи по теоретической механике. И.В. Мещерский. СПб.: Издательство «Лань», 2001. – 448 с.)

Определить период свободных колебаний груза массой m , зажатого между двумя пружинами с коэффициентами жесткости c_1 и c_2 .

Для решения задачи использовать следующие значения параметров:

$$m = 10 \text{ кг}, c_1 = 300 \text{ Н/м}, c_2 = 500 \text{ Н/м}$$




Точный теоретический ответ

Период определяется по следующей формуле:

$$T = 2\pi \sqrt{\frac{m}{c_1 + c_2}}.$$

Решение задачи в EULER



Система состоит из двух звеньев.

- Инерциальное звено (body1). В проекте звено не отображается.
- Груз (body2). Звено отображается сферой solid1 и имеет массу m .

Пружины моделируются силовыми элементами force1 и force2 типа «поступательная пружина с коэффициентом жесткости». Для того чтобы груз начал совершать колебания, ему сообщается начальная скорость с помощью объекта «условие состояния механизма» condition1. Для определения смещения и скорости груза используются соответственно датчики s и v .

Период колебаний определяется как промежуток времени, за которое груз возвращается в первоначальное положение и при этом имеет положительную скорость (как в начальный момент). Для определения периода создано событие event1. Оно останавливает расчет, когда груз проходит первоначальное положение и при этом выполняются условия: время больше и груз имеет положительную скорость. В момент

остановки расчета датчик time будет иметь значение равное периоду свободных колебаний груза.

Для решения задачи используется команда command1 «расчет динамики движения» с временем интегрирования 1 [s] и заданным постоянным шагом.

Результаты моделирования

Относительное отличие решения задачи в EULER в зависимости от шага численного интегрирования (использовался постоянный шаг интегрирования) представлено в следующей таблице.

Шаг интегрирования [s]	Относительное отличие от теоретического решения T_delta_rel
0.1	0.003 4
0.01	0.000 001 2
0.001	0.000 001 4
0.0001	0.000 000 79

Текст проекта в EULER

```

scalar m=10 [kg];
scalar c1=300 [ kg/ s2 ];
scalar c2=500 [ kg/ s2 ];
point point1=point( 0 [ m ], 1 [ m ], 0 [ m ] );
point point2=point( 0 [ m ], -1 [ m ], 0 [ m ] );
body body1=body( color = RGB( 229, 229, 229 ) );
set ground = body1;
point point3=point( 0 [ m ], 0 [ m ], 0 [ m ] );
solid solid1=sphere( point3, 0.1 [ m ], mass = m );
body body2=body( color = RGB( 255, 153, 153 ) );
body body2 < ( solid1 );
force force1=spring( body1, point1, body2, point3, c1 );
force force2=spring( body1, point2, body2, point3, c2 );
sensor s=bodyDisplacement( body1, point3, projectY, body2, point3 );
scalar v0=1 [ m/ s ];
condition condition1=transVelocity( body1, projectY, body2, point3, v0 );
sensor v=transVelocity( body1, projectY, body2, point3 );
event event1=reformsBySensor( list( stop( ) ), s, 0 [ m ], logCondition = list( ( " time > 0 [ s ] ) & ( v > 0 [ m/ s ] )" ) );
sensor T_theoretical=2*PI*sqrt(m/(c1+c2));
sensor T_delta_rel=abs( ( time - T_theoretical ) / T_theoretical );
command command1=constRK4( 1.00000e+000 [ s ], 1.00000e-003 [ s ] );

```

Λ//

Λ Список главных команд;
set dynamics = command1;

Λ//

Λ Единицы измерения;
set units = SI;